
Работа со сводными 
таблицами

В ЭТОЙ ГЛАВЕ...

  Ознакомительный пример

  Создание сложных сводных таблиц

  Создание нескольких сводных таблиц

  Создание обратной сводной табли цы

Эта глава посвящена сводным таблицам Excel — одной из важнейших тем, которые 
просто невозможно обойти стороной.

Ознакомительный пример

Сводные таблицы являются одним из самых мощных инновационных средств Excel. 
Они появились в Excel 5 и поддерживаются только в Excel (пока ни один программный 
продукт для управления электронными таблицами не предоставляет подобной возмож-
ности). В этой главе отсутствует информация об основах использования сводных таблиц. 
Предполагается, что вы уже знакомы с данной темой (как и со способами создания и из-
менения сводных таблиц вручную).

Создав сводную таблицу на основе базы данных или списка, можно быстро отобразить 
необходимые итоговые данные, которые другим способом представить на экране или бу-
маге просто невозможно. Кроме того, такое представление данных можно получить очень 
быстро и безо всяких формул. В Excel также поддерживается возможность создания кода 
VBA, который позволяет управлять сводными таблицами.

Этот раздел начинается с простого примера создания сводной таблицы средствами VBA.
На рис. 15.1 показана простая база данных рабочего листа (диапазон ячеек рабочего 

листа). Она включает следующие четыре поля: Торговый представитель, Регион, Ме-
сяц и Продажи. Каждая запись содержит сведения об объеме продаж торгового предста-
вителя за определенный месяц.

Рассматриваемый далее пример находится на сайте книги (файл simple pivot 
table.xlsm).

15

Рис. 15.1. Эта таблица является хорошим кандидатом в сводные таблицы

Создание сводной таблицы

 На рис. 15.2 показана сводная таблица, которая создана на основе указанных данных. 
Она обеспечивает отображение суммарного объема ежемесячных продаж каждого торго-
вого представителя и содержит следующие поля.

  Регион. Поле фильтра отчета в сводной таблице.
  Торговый представитель. Поле строки в сводной таблице.
  Месяц. Поле столбца в сводной таблице.
  Продажи. Поле данных в сводной таблице, которое использует функцию Сумма (Sum).

Рис. 15.2. Сводная таблица, созданная на основе таблицы, показанной на рис. 15.1

Перед созданием сводной таблицы была включена функция записи макроса. Созданная 
сводная таблица размещается на новом рабочем листе. Далее представлен автоматически 
сгенерированный код макроса.
Sub CreatePivotTable()
   Sheets.Add
   ActiveWorkbook.PivotCaches.Create _
      (SourceType:=xlDatabase, _
      SourceData:="Лист1!R1C1:R13C4", _
      Version:=xlPivotTableVersion15).CreatePivotTable _
      TableDestination:="Лист2!R3C1", _
      TableName:="PivotTable1", _
      DefaultVersion:=xlPivotTableVersion15)
   Sheets("Лист2").Select
   Cells(3, 1).Select
   With ActiveSheet.PivotTables("PivotTable1") _
      .PivotFields("Регион")
      .Orientation = xlPageField
      .Position = 1
   End With
   With ActiveSheet.PivotTables("PivotTable1") _
      .PivotFields("Торговый представитель")
      .Orientation = xlRowField
      .Position = 1
   End With
   With ActiveSheet.PivotTables("PivotTable1") _
      .PivotFields("Месяц")
      .Orientation = xlColumnField
      .Position = 1
   End With
   ActiveSheet.PivotTables("PivotTable1") _
      .AddDataField ActiveSheet.PivotTables("PivotTable1") _
      .PivotFields("Продажи"), "Сумма по полю Продажи", xlSum
End Sub

Запуск записанного макроса скорее всего приведет к ошибке. В процессе проверки 
кода вы обнаружите, что функция записи макроса “жестко закодировала” название ра-
бочего листа (Лист2), на котором создается сводная таблица. Если подобный лист уже 
существует (либо добавляется новый лист с другим именем), выполнение макроса при-
ведет к ошибке.

Несмотря на неработоспособность записанного макроса, вряд ли его можно назвать 
совсем бесполезным. Разработчики кода VBA для сводных таблиц найдут в этом коде не-
мало полезного для себя.

Просмотр кода VBA

Код VBA, записанный при создании сводной таблицы, может привести вас в замеша-
тельство. Для того чтобы разобраться в записанном макросе, потребуется кое-что знать 
об используемых объектах (соответствующая информация содержится в интерактивном 
справочном руководстве).

  PivotCaches — коллекция объектов PivotCache в объекте Workbook (данные 
для сводной таблицы, которые хранятся в кеше сводной таблицы). 

  PivotTables — коллекция объектов PivotTable в объекте Worksheet.
  PivotFields — коллекция полей в объекте PivotTable.
  PivotItems — коллекция отдельных элементов данных в поле.
  CreatePivotTable — метод объекта PivotCache, который создает сводную та-

блицу на основе данных, содержащихся в кеш-памяти.

ВЫБОР ДАННЫХ ДЛЯ СВОДНОЙ ТАБЛИЦЫ

Данные, необходимые для  создания сводной таблицы, должны быть представлены 
в  виде прямоугольной базы данных (прямоугольный диапазон ячеек сводной табли-
цы). Причем база данных может храниться как в виде диапазона листов (таблица либо 
обычный диапазон), так и в файле внешней базы данных. Несмотря на то что Excel мо-
жет генерировать сводные таблицы на основе практически любой базы данных, долж-
ны быть соблюдены определенные условия.
Поля в таблице базы данных, на основе которой создается сводная таблица, относятся 
к следующим двум типам.

  Данные (Data). Включает суммируемые значения. Например, в случае с базой 
данных о продажах поле Продажи является полем данных.

  Категория (Category). Описывает данные. Например, в  случае с  той же базой 
данных о  продажах поля Торговый представитель, Регион и Месяц  — это 
примеры полей категорий, описывающих данные в поле Продажи.

Таблица базы данных, на основе которой может быть построена сводная таблица, долж-
на быть нормализована. Иными словами, каждая запись (или строка) должна включать 
информацию, описывающую данные.
Единственная таблица базы данных может включать произвольное количество полей 
данных и категорий. При создании сводной таблицы обычно выполняется суммирова-
ние по одному или нескольким полям данных. Ну а значения в полях категорий отобра-
жаются в сводной таблице в виде строк, столбцов либо фильтров.
Если изложенная выше концепция не вполне понятна, обратитесь к  находящемуся 
на  сайте книги файлу normalized data.xlsx. Эта рабочая книга содержит пример 
диапазона данных до  и после выполнения нормализации, являющейся необходимым 
условием построения сводной таблицы.

Усовершенствование записанного кода сводной таблицы

Как и в случае с большинством записанных макросов, предыдущий пример не настолько 
эффективен, как следовало бы ожидать. Как уже отмечалось, его выполнение может завер-
шиться ошибкой. В принципе, его желательно упростить, чтобы сделать немного понятнее, 
а также исключить возможность появления ошибок. Ниже приведен код, переписанный 
вручную, который создает ту же сводную таблицу, что и макрос из предыдущего раздела.
Sub CreatePivotTable()
   Dim PTCache As PivotCache
   Dim PT As PivotTable

'  Создание области кэша
   Set PTCache = ActiveWorkbook.PivotCaches.Create( _
      SourceType:=xlDatabase, _
      SourceData:=Range("A1").CurrentRegion)

'  Добавление нового листа в сводную таблицу
   Worksheets.Add

'  Создание сводной таблицы
   Set PT = ActiveSheet.PivotTables.Add( _
      PivotCache:=PTCache, _
      TableDestination:=Range("A3"))

'  Определение полей
   With PT
      .PivotFields("Регион").Orientation = xlPageField
      .PivotFields("Месяц").Orientation = xlColumnField
      .PivotFields("Торговый представитель") _
         .Orientation = xlRowField
      .PivotFields("Продажи").Orientation = xlDataField

      'заголовки полей отсутствуют
      .DisplayFieldCaptions = False
   End With
End Sub

В данном случае процедура CreatePivotTable была упрощена (что облегчило ее по-
нимание) благодаря объявлению двух переменных объекта: PTCache и PT. Новый объект 
PivotCache был создан с помощью метода Create. Также был добавлен рабочий лист, 
который стал активным (на этом листе размещается сводная таблица). Затем был создан 
объект PivotTable с помощью метода Add из коллекции PivotTables. В последнем 
фрагменте кода добавляются поля в сводную таблицу, а также задается их положение в та-
блице путем присвоения значения свойству Orientation.

Обратите внимание на то, что исходный макрос жестко привязан к диапазону данных, 
на основе которого создается объект PivotCache ('Лист1!R1C1:R13C4'), и к местопо-
ложению сводной таблицы (Лист2). В процедуре CreatePivotTable сводная таблица 
основана на текущем диапазоне, окружающем ячейку A1. Это гарантирует, что макрос бу-
дет выполняться даже тогда, когда в диапазон добавлены дополнительные данные.

Добавление рабочего листа до того, как была создана сводная таблица, исключает не-
обходимость жесткого кодирования ссылки на лист. Еще одно отличие заключается в том, 
что написанный вручную макрос не определяет имя сводной таблицы. Да это и не требует-
ся, поскольку используется переменная объекта PT, которая выполняет эту задачу.

Данный код можно сделать более универсальным, используя индексы вместо 
названий в коллекции PivotFields. При этом, если пользователь изменяет за-
головки столбцов, код остается работоспособным. Например, в  более универ-
сальном макросе используется оператор PivotFields(1) вместо оператора 
PivotFields('Регион').

СОВМЕСТИМОСТЬ СВОДНЫХ ТАБЛИЦ

Если планируется организовать общий доступ к  рабочим книгам, включающим свод-
ные таблицы, со стороны пользователей предыдущих версий Excel, уделите осо-
бое внимание вопросам совместимости. При просмотре кода записанного макроса 
(см. раздел “Создание сводной таблицы”) нетрудно заметить следующую инструкцию:

DefaultVersion:=xlPivotTableVersion15

Если рабочая книга находится в  режиме совместимости, на  ее месте будет такая ин-
струкция:

DefaultVersion:=xlPivotTableVersion10

Вы также увидите, что код записанного макроса совершенно отличается. Причина 
заключается в  существенных изменениях в  сводных таблицах, начиная с  версии 
Excel 2007.
Предположим, что в  среде Excel  2013 создана рабочая книга, которая была передана 
пользователям Excel 2003. Они увидят таблицу, но не смогут ее обновить. Другими сло-
вами, они получат в свое распоряжение статическую таблицу, включающую ряд чисел.
Для обеспечения обратной совместимости сводных таблиц в Excel 2013 следует сначала 
сохранять файлы в формате XLS, а затем повторно их открывать. В этом случае обеспе-
чивается работоспособность сводных таблиц при открытии содержащих их документов 
в версиях, предшествующих Excel 2007. Но в этом случае вы не сможете воспользоваться 
новыми средствами сводных таблиц, которые появились в современных версиях Excel.
К счастью, существует модуль проверки совместимости Excel, который поможет выявить 
проблемы, связанные с совместимостью (см. следующий рисунок). Но этот модуль не в со-
стоянии проверить поддержку обратной совместимости кода, генерируемого макросом.

Рассматриваемый в этой главе макрос не в состоянии генерировать сводные таблицы, 
поддерживающие обратную совместимость.

Чтобы лучше понять рассматриваемую тему, запишите собственный макрос и изучите 
ключевые объекты, методы и свойства. После этого обратитесь к разделам справочного 
руководства, чтобы разобраться, как все это работает. Практически во всех случаях необ-
ходимо вносить изменения в записанный макрос. Как только вам станут понятны прин-
ципы управления сводными таблицами, можете приступать к созданию кода без предва-
рительной записи макроса.

Создание сложных сводных таблиц

В этом разделе будет представлен код VBA, используемый для создания сравнительно 
сложной сводной таблицы.

На рис. 15.3 показан фрагмент базы данных на листе. В этой таблице содержится 15840 
строк иерархически упорядоченной информации о бюджете компании. В компании суще-
ствует пять подразделений; каждое подразделение включает одиннадцать отделов. Отдел 
имеет четыре бюджетные категории, а каждая категория состоит из нескольких пунктов. 
Бюджетные и фактические расходы указываются для каждого (из двенадцати) месяцев. 
Цель — создать сводную таблицу, которая суммирует эту информацию.

Рис. 15.3. Данные из этой рабочей книги будут собраны в сводной таблице

Рассматриваемая в данном разделе рабочая книга находится на сайте книги (файл 
budget pivot table.xlsm).

На рис. 15.4 показана сводная таблица, созданная на основе приведенных выше дан-
ных. Обратите внимание на то, что она содержит вычисляемое поле, которое называется 
Отклонение. Значение этого поля представляет собой разницу между значениями полей 
План и Факт.

Можно также вставить новую колонку в таблицу и создать формулу, вычисляющую 
разницу между плановыми и фактическими показателями. Эта возможность будет 
недоступной в том случае, если данные берутся из внешнего источника.

Excel 2013 Power Programming with VBA.indb   545Excel 2013 Power Programming with VBA.indb   545 08.04.2014   13:08:2808.04.2014   13:08:28



Часть IV.  Профессиональные методики программирования546

Рис. 15.4. Сводная таблица, созданная на основе данных из рис. 15.3

Код сводной таблицы

Ниже приведен код, генерирующий сводную таблицу.
Sub CreatePivotTable()
   Dim PTcache As PivotCache
   Dim PT As PivotTable

   Application.ScreenUpdating = False
'  Удаление листа сводной таблицы (при его наличии)
   On Error Resume Next
   Application.DisplayAlerts = False
   Sheets("Сводная таблица").Delete
 On Error GoTo 0

'  Создание кеша сводной таблицы 
   Set PTcache = ActiveWorkbook.PivotCaches.Create( _
      SourceType:=xlDatabase, _
      SourceData:=Range("A1").CurrentRegion.Address)

'  Добавление нового рабочего листа
   Worksheets.Add
   ActiveSheet.Name = "Сводная таблица"
   ActiveWindow.DisplayGridlines = False

'  Создание сводной таблицы на основе данных кеша
   Set PT = ActiveSheet.PivotTables.Add( _
      PivotCache:=PTcache, _
      TableDestination:=Range("A1"), _
      TableName:="BudgetPivot")

   With PT
'     Добавление полей
      .PivotFields("Категория").Orientation = xlPageField
      .PivotFields("Подразделение").Orientation = xlPageField
      .PivotFields("Отдел").Orientation = xlRowField
      .PivotFields("Месяц").Orientation = xlColumnField
      .PivotFields("План").Orientation = xlDataField
      .PivotFields("Факт").Orientation = xlDataField
      .DataPivotField.Orientation = xlRowField

'     Добавление вычисляемого поля, определяющего отклонение
      .CalculatedFields.Add "Отклонение", "=План-Факт"
      .PivotFields("Отклонение").Orientation = xlDataField

'     Определение числового формата
      .DataBodyRange.NumberFormat = "0,000"

'     Применение стиля
      .TableStyle2 = "PivotStyleMedium2"

'     Сокрытие заголовков полей
      .DisplayFieldCaptions = False

'     Изменение заголовков
      .PivotFields("Сумма по полю План").Caption = " План"
      .PivotFields("Сумма по полю Факт").Caption = " Факт"
      .PivotFields("Сумма по полю Отклонение").Caption = _
         " Отклонение"
   End With
End Sub

Принцип работы сводной таблицы

Процедура CreatePivotTable начинает свою работу с удаления листа Сводная 
таб лица, если он существует. После этого создается объект PivotCache, добавляется но-
вый лист Сводная таблица и создается сводная таблица на основе объекта PivotCache. 
Далее программа добавляет следующие поля к созданной сводной таблице:

  Категория — поле фильтра отчета (страницы).
  Подразделение — поле фильтра отчета (страницы).
  Отдел — поле строки.
  Месяц — поле столбца.
  План — поле данных.
  Факт — поле данных.

Обратите внимание на то, что свойство Orientation объекта DataPivotField уста-
новлено равным xlRowField с помощью следующего оператора:
.DataPivotField.Orientation = xlRowField

Этот оператор определяет общую ориентацию сводной таблицы и представляет поле в об-
ласти ЗНАЧЕНИЯ в списке полей сводной таблицы (рис. 15.5). Попытайтесь переместить это 
поле в область КОЛОННЫ, после чего посмотрите, как изменится макет сводной таблицы.

Затем процедура использует метод Add из коллекции CalculatedFields для созда-
ния вычисляемого поля Отклонение, величина которого представляет собой результат 
вычитания значения поля Факт из значения поля План. Это вычисляемое поле относится 
к категории полей данных.

Чтобы добавить вычисляемое поле в сводную таблицу вручную, воспользуйтесь 
командой Работа со сводными таблицамиАнализВычисленияПоля, эле-
менты и наборыВычисляемое поле (PivotTable ToolsAnalysisCalculations
Fields, Items & SetsCalculation Field) для перехода в диалоговое окно Вставка вы-
числяемого поля (Insert Calculated Field).

Рис. 15.5. Список полей сводной таблицы
Также код выполняет небольшие “косметические” улучшения сводной таблицы.

  Применяет числовой формат к объекту DataBodyRange (этот объект представляет 
данные всей сводной таблицы).

  Применяет стиль.
  Скрывает заголовки, что эквивалентно команде Работа со сводными таблицами
АнализПоказать/СкрытьЗаголовки полей (PivotTable ToosAnalysisShow/
HideField Headers).

  Изменяет заголовки, отображенные в сводной таблице. Например, название Сумма 
по полю План (Sum of Budget) заменяется названием План (Budget). Обратите 
внимание на то, что перед названием План вставляется пробел. В Excel не допуска-
ется изменение заголовка, который соответствует имени поля, но, как видите, это 
ограничение можно обойти, добавив пробел.

В процессе создания процедуры я постоянно пользовался средством записи ма-
кросов, чтобы изучить различные свойств. Также мне очень помогла интерактивная 
справочная система (ну и многие вещи пришлось делать методом проб и ошибок).

Создание нескольких сводных таблиц

При выполнении данного упражнения создается целая серия сводных таблиц, в ко-
торых суммируются данные, собранные в процессе опроса покупателей. Эти данные на-
ходятся в базе данных рабочих листов (рис. 15.6), которая состоит из 150 строк. Каждая 

строка включает указание пола респондента, а также числовой рейтинг, который изменя-
ется от 1 до 5 по каждому из 14 пунктов опроса.

Рис. 15.6. Создание набора сводных таблиц, суммирующих данные опроса

Эта рабочая книга находится на сайте книги (файл survey data pivot tables.xlsm).

На рис. 15.7 показаны некоторые из 28 сводных таблиц, созданные макросом. Сумми-
рование по каждому элементу опроса производилось с помощью двух сводных таблиц (в 
одной отображаются фактические результаты опроса, в другой — процент респондентов, 
ответивших на пункты опроса).

Рис. 15.7. Некоторые из 28 сводных таблиц, созданных с помощью процедуры VBA

Ниже приводится процедура VBA, с помощью которой создавались сводные таблицы.
Sub MakePivotTables()
'  Эта процедура создает 28 сводных таблиц
   Dim PTCache As PivotCache
   Dim PT As PivotTable
   Dim SummarySheet As Worksheet
   Dim ItemName As String
   Dim Row As Long, Col As Long, i As Long

   Application.ScreenUpdating = False

'  Удаление листа итогов при его наличии
   On Error Resume Next
   Application.DisplayAlerts = False
   Sheets("Итог").Delete
   On Error GoTo 0

'  Добавление листа итогов
   Set SummarySheet = Worksheets.Add
   ActiveSheet.Name = "Итог"

'  Создание кеша сводной таблицы
   Set PTCache = ActiveWorkbook.PivotCaches.Create( _
      SourceType:=xlDatabase, _
      SourceData:=Sheets("ДанныеОпроса").Range("A1"). _
      CurrentRegion)

   Row = 1
      For i = 1 To 14
         For Col = 1 To 6 Step 5 '2 колонки
            ItemName = Sheets("ДанныеОпроса").Cells(1, i + 2)
            With Cells(Row, Col)
               .Value = ItemName
               .Font.Size = 16
            End With

'           Создание сводной таблицы
            Set PT = ActiveSheet.PivotTables.Add( _
               PivotCache:=PTCache, _
            TableDestination:=SummarySheet.Cells( _
               Row + 1, Col))

'           Добавление полей
            If Col = 1 Then 'таблицы частот
               With PT.PivotFields(ItemName)
                  .Orientation = xlDataField
                  .Name = "Частота"
                  .Function = xlCount
               End With
            Else ' процентные таблицы
               With PT.PivotFields(ItemName)
                  .Orientation = xlDataField
                  .Name = "Процент"
                  .Function = xlCount

                  .Calculation = xlPercentOfColumn
                  .NumberFormat = "0,0%"
               End With
            End If
            PT.PivotFields(ItemName).Orientation = xlRowField
            PT.PivotFields("Пол").Orientation = xlColumnField
            PT.TableStyle2 = "PivotStyleMedium2"
            PT.DisplayFieldCaptions = False
            If Col = 6 Then
'              добавление шкал данных в последнюю колонку
               PT.ColumnGrand = False
               PT.DataBodyRange.Columns(3).FormatConditions. _
                  AddDatabar
               With pt.DataBodyRange.Columns(3). _
                  FormatConditions(1)
                  .BarFillType = xlDataBarFillSolid
                  .MinPoint.Modify newtype:= _
                      xlConditionValueNumber, newvalue:=0
                  .MaxPoint.Modify newtype:= _
                     xlConditionValueNumber, newvalue:=1
               End With
            End If
         Next Col
         Row = Row + 10
      Next i

'     Замена чисел описательным текстом
      With Range("A:A,F:F")
         .Replace "1", "Решительно не согласен"
         .Replace "2", "Не согласен"
         .Replace "3", "Колеблюсь"
         .Replace "4", "Согласен"
         .Replace "5", "Полностью согласен"
      End With
End Sub

Обратите внимание на то, что все сводные таблицы были созданы на основе единствен-
ного объекта PivotCache.

Сводные таблицы создаются с помощью вложенных циклов. Значение счетчика цикла 
Col изменяется от 1 до 6 с помощью параметра Step. Для второго столбца сводных та-
блиц немного изменяется характер обработки. Выполняются следующие действия:

  отображается величина (в виде процента от значения в столбце);
  не отображаются окончательные итоги для строк;
  используется числовой формат;
  отображаются цветовые шкалы, с помощью которых реализуется условное форма-

тирование.

Переменная Row отслеживает начальную строку в каждой сводной таблице. Завершаю-
щая операция заключается в замене числовых категорий в столбцах A и F текстом. Напри-
мер, вместо единицы подставляется текст “Решительно не согласен”.

Создание обратной сводной таблицы

 Сводная таблица представляет собой результат суммирования данных в обычной таб-
лице. А как быть в случае, если у вас имеется итоговая таблица, на основе которой нужно 
воссоздать исходную таблицу? Соответствующий пример приводится на рис. 15.8. Диапа-
зон ячеек B2:F14 представляет собой итоговую таблицу — упрощенный вариант сводной 
таблицы. В столбцах I:K находится состоящая из 48 строк таблица, созданная на основе 
итоговой таблицы. В этой таблице каждая строка содержит точку данных, а первые два 
столбца включают описание этой точки.

Рис. 15.8. Итоговая таблица (слева) может быть преобразована в обычную (справа)

В Excel не существует прямого способа преобразования итоговой таблицы в обычную, 
но эту работу может выполнить макрос VBA. После его создания я разработал диалоговое 
окно UserForm, показанное на рис. 15.9. В нем определяются входной и выходной диапазо-
ны, а также имеется опция преобразования выходного диапазона в таблицу.

Эта рабочая книга находится на сайте книги (файл reverse pivot table.xlsm).

После щелчка мышью на кнопке OK в диалоговом окне UserForm код VBA проверяет 
диапазоны и вызывает процедуру ReversePivot с помощью следующего оператора:
Call ReversePivot(SummaryTable, OutputRange, cbCreateTable)

Рис. 15.9. В этом диалоговом окне пользователь может определить диапазоны

Этой процедуре передаются следующие три аргумента.

  SummaryTable. Объект Range, который представляет итоговую таблицу.
  OutputRange. Объект Range, представляющий левую верхнюю ячейку выходного 

диапазона.
  cbCreateTable. Флажок (объект Checkbox) в окне UserForm.

Эта процедура работает с итоговой таблицей любого размера. Количество строк дан-
ных в выходной таблице можно подсчитать по формуле (r-1) * (c-1), в которой пере-
менные r и c представляют собой количество строк и столбцов в итоговой таблице соот-
ветственно.

Ниже приведен код процедуры ReversePivot.
Sub ReversePivot(SummaryTable As Range, _
   OutputRange As Range, CreateTable As Boolean)
   Dim r As Long, c As Long
   Dim OutRow As Long, OutCol As Long

'  Преобразование диапазона
   OutRow = 2
   Application.ScreenUpdating = False
   OutputRange.Range("A1:C3") = Array("Столбец", _
      "Столбец2", "Столбец3")
   For r = 2 To SummaryTable.Rows.Count

      For c = 2 To SummaryTable.Columns.Count
         OutputRange.Cells(OutRow, 1) = SummaryTable.Cells(r, 1)
         OutputRange.Cells(OutRow, 2) = SummaryTable.Cells(1, c)
         OutputRange.Cells(OutRow, 3) = SummaryTable.Cells(r, c)
         OutRow = OutRow + 1
      Next c
   Next r

'  Создать таблицу?
   If CreateTable Then _
      ActiveSheet.ListObjects.Add xlSrcRange, _
         OutputRange.CurrentRegion, , xlYes
End Sub

Описанная процедура достаточно проста. Код выполняет циклический обход строк 
и столбцов во входном диапазоне, а затем записывает данные в выходной диапазон, ко-
торый всегда состоит из трех столбцов. Переменная OutRow отслеживает текущую стро-
ку в выходном диапазоне. Если пользователь устанавливает флажок, выходной диапазон 
преобразуется в таблицу. При этом используется метод Add из коллекции ListObjects.

